Gisser and Sánchez results

This program replicates the computations of Gisser and Sánchez (*WRR* 16, August 1980, pp. 638-642) for examining the pumping cost externality. The G&S work pertains to a simple situation of linear ag demand for water and pumping costs which depend linearly on water table height. Getting this program to work (in *Mathematica* 4.2) properly required a little finessing as my installation of DSolve, the differential equation solver, appeared to have precision problems for the originally given problem specification. Entering the parameters as exact (integer) numbers fixes things. Be patient.

■ Preliminaries

```
Off[General::spell];
Off[General::spell1];
thinn = AbsoluteThickness[.5];
medum = AbsoluteThickness[1.];
thick = AbsoluteThickness[1.5];
$TextStyle = {FontFamily → "Helvetica", FontSlant -> "Plain", FontSize → 9};
SetOptions[Plot, PlotPoints → 40, ImageSize → 384];
SetOptions[Plot, FrameStyle -> medum, AxesStyle -> medum, PlotStyle → medum];
SetOptions[ListPlot, AxesStyle -> medum, PlotStyle → medum, ImageSize → 384];
SetOptions[ParametricPlot, PlotPoints → 40,
 FrameStyle -> medum, AxesStyle -> medum, PlotStyle → medum];
black = GrayLevel[0];
BGray = GrayLevel[0.3];
WGray = GrayLevel[0.6];
<< Graphics`TickControl`
```

G&S Table 1 Data and Discount rate

```
g = 470365;
k = -3259;
C0 = 125;
C1 = -35 / 1000;
α = 27 / 100;
AS = 135000;
R = 173000;
H0 = 3400;
r = 1 / 100;
```

Uncontrolled (Competition) Replication

■ Solve eq. 7

"u" prefix stands for uncontrolled (no control, same as competition) results

```
DSolve[{AS * uh′[t] == R + (α - 1.) * (g + k C0) + (α - 1.) * k * C1 * uh[t], uh[0] == H0}, uh[t], t]
```

$\{\{uh[t] \to (1874.59 + 1525.41\ 2.71828^{0.000616796\ t})\ e^{-0.000616796\ t}\}\}$

```
uh = Simplify[uh[t] /. Flatten[%]]
```

$1525.41 + 1874.59\ e^{-0.000616796\ t}$

```
uw = Simplify[g + k C0 + k C1 uh]
```

$236986. + 213825.\ e^{-0.000616796\ t}$

Optimal Control Replication (discount rate = 1%)

▪ P. 640 Variable Definitions (can't use N as a variable like G&S; use η instead)

$$m = \frac{\alpha - 1}{AS};$$
$$n = r\, k\, C_1;$$
$$M = \frac{R}{AS};$$
$$\eta = -g\, r - C_0\, k\, r + \frac{k\, C_1\, R}{AS};$$

■ Solve G&S eqs 18 & 19 Simultaneously

```
rslt = DSolve[{w'[t] == r*w[t] - n*h[t] + η, h'[t] == m*w[t] + M, h[0] == H₀}, {w[t], h[t]}, t]
```

$$\Bigg\{\Bigg\{w[t] \to \frac{1}{11057768586}\Bigg(e^{-\frac{(4500+\sqrt{25246047})\,t}{900000}}\Bigg(1310269839300000\, e^{\frac{(4500+\sqrt{25246047})\,t}{900000}} - 138207247700\sqrt{25246047}\, e^{\frac{(4500+\sqrt{25246047})\,t}{900000}} - 392637943300\sqrt{25246047}\, e^{\frac{(4500+\sqrt{25246047})\,t}{450000}} + 1310269839300000\, e^{\frac{\sqrt{\frac{8415349}{3}}\,t}{150000}+\frac{(4500-\sqrt{25246047})\,t}{900000}} + 138207247700\sqrt{25246047}\, e^{\frac{\sqrt{\frac{8415349}{3}}\,t}{150000}+\frac{(4500-\sqrt{25246047})\,t}{900000}} + 392637943300\sqrt{25246047}\, e^{\frac{(4500-\sqrt{25246047})\,t}{900000}+\frac{(4500+\sqrt{25246047})\,t}{900000}} + 5528884293\, e^{\frac{(4500+\sqrt{25246047})\,t}{450000}}\, C[1] + 985500\sqrt{25246047}\, e^{\frac{(4500+\sqrt{25246047})\,t}{450000}}\, C[1] + 5528884293\, e^{\frac{(4500-\sqrt{25246047})\,t}{900000}+\frac{(4500+\sqrt{25246047})\,t}{900000}}\, C[1] - 985500\sqrt{25246047}\, e^{\frac{(4500-\sqrt{25246047})\,t}{900000}+\frac{(4500+\sqrt{25246047})\,t}{900000}}\, C[1]\Bigg)\Bigg),$$

$$h[t] \to -\frac{1}{3783913121286270}\Bigg(e^{-\frac{(4500+\sqrt{25246047})\,t}{900000}}\Bigg(-3128466982674947300\, e^{\frac{(4500+\sqrt{25246047})\,t}{900000}} + 644067484650000\sqrt{25246047}\, e^{\frac{(4500+\sqrt{25246047})\,t}{900000}} - 3304185323511711700\, e^{\frac{(4500+\sqrt{25246047})\,t}{450000}} + 588956914950000\sqrt{25246047}\, e^{\frac{(4500+\sqrt{25246047})\,t}{450000}} - 3128466982674947300\, e^{\frac{\sqrt{\frac{8415349}{3}}\,t}{150000}+\frac{(4500-\sqrt{25246047})\,t}{900000}} - 644067484650000\sqrt{25246047}\, e^{\frac{\sqrt{\frac{8415349}{3}}\,t}{150000}+\frac{(4500-\sqrt{25246047})\,t}{900000}} - 3304185323511711700\, e^{\frac{(4500-\sqrt{25246047})\,t}{900000}+\frac{(4500+\sqrt{25246047})\,t}{900000}} - 588956914950000\sqrt{25246047}\, e^{\frac{(4500-\sqrt{25246047})\,t}{900000}+\frac{(4500+\sqrt{25246047})\,t}{900000}} + 364711431\sqrt{25246047}\, e^{\frac{(4500+\sqrt{25246047})\,t}{450000}}\, C[1] - 364711431\sqrt{25246047}\, e^{\frac{(4500-\sqrt{25246047})\,t}{900000}+\frac{(4500+\sqrt{25246047})\,t}{900000}}\, C[1]\Bigg)\Bigg)\Bigg\}\Bigg\}$$

■ Install these Solutions as w* and h*

```
h* = N[Expand[h[t] /. Flatten[rslt]]]
```

$$-28.457 + 1655.28\ 2.71828^{-0.000582827\,t} + 1682.02\ 2.71828^{0.\,t} + 91.161\ 2.71828^{0.0105828\,t} + 0.000484289\ 2.71828^{-0.000582827\,t}\, C[1] - 0.000484289\ 2.71828^{0.0105828\,t}\, C[1]$$

```
w* = N[Expand[w[t] /. Flatten[rslt]]]
```

$$55693.1 + 178411.\ 2.71828^{-0.000582827\,t} + 181293.\ 2.71828^{0.\,t} - 178411.\ 2.71828^{0.0105828\,t} + 0.0521982\ 2.71828^{-0.000582827\,t}\, C[1] + 0.947802\ 2.71828^{0.0105828\,t}\, C[1]$$

Observe the 2.71828 terms in the above solutions; that's "*e*".

```
h* /. t → 0
```

```
3400. + 0. C[1]
```

```
w* /. t → 0
```

```
236986. + 1. C[1]
```

▪ Use Transversality Condition to Resolve Constant

$$\lambda = N\left[\text{Expand}\left[\frac{(AS)\; e^{-r\,t}\left(\frac{w^*}{k} - \frac{g}{k} - C_0 - C_1\, h^*\right)}{\alpha - 1}\right]\right]$$

$$-590049.\ 2.71828^{-0.0105828\,t} - 829452.\ 2.71828^{-0.01\,t} - 1.0714\times10^{7}\ 2.71828^{0.000582827\,t} - 0.172632\ 2.71828^{-0.0105828\,t}\ C[1] + 56.9175\ 2.71828^{0.000582827\,t}\ C[1]$$

```
Solve[Limit[λ, t → ∞] == 0, C[1]]
```

```
{{C[1] → Indeterminate}}
```

```
transv = Solve[λ == 0., C[1]]
```

$$\left\{\left\{C[1] \to \frac{0. + 590049.\ 2.71828^{-0.0105828\,t} + 829452.\ 2.71828^{-0.01\,t} + 1.0714\times10^{7}\ 2.71828^{0.000582827\,t}}{-0.172632\ 2.71828^{-0.0105828\,t} + 56.9175\ 2.71828^{0.000582827\,t}}\right\}\right\}$$

```
Unprotect[C]
```

```
{C}
```

```
C[1] = C[1] /. Flatten[transv]
```

$$\frac{0. + 590049.\ 2.71828^{-0.0105828\,t} + 829452.\ 2.71828^{-0.01\,t} + 1.0714\times10^{7}\ 2.71828^{0.000582827\,t}}{-0.172632\ 2.71828^{-0.0105828\,t} + 56.9175\ 2.71828^{0.000582827\,t}}$$

```
limC = Limit[C[1], t → ∞]
```

```
188237.
```

The very last command took a while to run, but it's what we want.

▪ Install Constant and Request Complete Solution

Install the solution for C[1]:

```
C[1] = limC
```

```
188237.
```

Now that the constant C is set, what does the solution look like?

```
w*
h*
```

$55693.1 + 188237.\ 2.71828^{-0.000582827\ t} + 181293.\ 2.71828^{0.\ t} + 0.\ 2.71828^{0.0105828\ t}$

$-28.457 + 1746.44\ 2.71828^{-0.000582827\ t} + 1682.02\ 2.71828^{0.\ t} + 4.26326 \times 10^{-14}\ 2.71828^{0.0105828\ t}$

The right-most terms are very, very small (numerical precision issue). Let's dump them.

```
w* = Chop[w*]
h* = Chop[h*]
```

$236986. + 188237.\ 2.71828^{-0.000582827\ t}$

$1653.56 + 1746.44\ 2.71828^{-0.000582827\ t}$

Graphing

■ In Preparation for graphing, what do things look like?

```
w* /. t → 0
h* /. t → 0
```

425223.

3400.

```
uw /. t → 0
uh /. t → 0
```

450811.

3400.

```
λ
```

$-622544.\ 2.71828^{-0.0105828\ t} - 829452.\ 2.71828^{-0.01\ t} + 0.\ 2.71828^{0.000582827\ t}$

$$\text{muc} = N\left[\frac{\lambda\,(\alpha - 1)}{AS}\right]$$

$-5.40741 \times 10^{-6}\ (-622544.\ 2.71828^{-0.0105828\ t} - 829452.\ 2.71828^{-0.01\ t} + 0.\ 2.71828^{0.000582827\ t})$

```
λ /. t → 0
muc /. t → 0
```

-1.452×10^{6}

7.85154

■ Plot Water Use and Head

```
plgw1 = Plot[{w*, uw}, {t, 0, 1000},
  PlotRange → {{0, 1050}, {100000, 460000}},
  Ticks → {TickTable[{0, 1000}, {200, 100}, MajorStyle → medum,
     MajorSize → 0.01, MinorStyle → thinn, MinorSize → 0.005],
    TickTable[{100000, 450000}, {50000, 50000}, MajorStyle → medum, MajorSize → 0.01]},
  AxesLabel → {"Year", "Ac-Ft"}]
```




```
- Graphics -
```

```
plgw2 = Plot[{h*, uh}, {t, 0, 1000},
  PlotRange → {{0, 1050}, {0, 3600}},
  Ticks → {TickTable[{0, 1000}, {200, 100}, MajorStyle → medum,
     MajorSize → 0.01, MinorStyle → thinn, MinorSize → 0.005],
    TickTable[{0, 3600}, {500, 250}, MajorStyle → medum, MajorSize → 0.01,
     MinorStyle → thinn, MinorSize → 0.005]},
  AxesLabel → {"Year", "Feet"}]
```




```
- Graphics -
```

▪ Plot Marginal User Cost

```
Plot[muc,{t,0,200},
     AxesLabel->{"Year","MUC"}]
```



- Graphics -

Resulting NPV (eq. 11)

$$npv4uw = \int_0^{\infty} e^{-r\,t} \left(\frac{1}{2\,k} * uw^2 - \frac{g}{k} * uw - (C_0 + C_1 * uh) * uw\right) dt$$

2.95063×10^9

$$npv4optw = \int_0^{\infty} e^{-r\,t} \left(\frac{1}{2\,k} * w^* ^2 - \frac{g}{k} * w^* - (C_0 + C_1 * h^*) * w^*\right) dt$$

2.96015×10^9

npv4uw / npv4optw

0.996783

$$npv4uwPart = \int_0^{20.} e^{-r*t} \left(\frac{1}{2\,k} * uw^2 - \frac{g}{k} * uw - (C_0 + C_1 * uh) * uw\right) dt$$

5.62018×10^8

$$npv4optwPart = \int_0^{20.} e^{-r*t} \left(\frac{1}{2\,k} * w^* ^2 - \frac{g}{k} * w^* - (C_0 + C_1 * h^*) * w^*\right) dt$$

5.60585×10^8

npv4uwPart / npv4optwPart

1.00256

Simple Alternative to Solving Transversality Condition to get C[1]

There are two e^{at} terms in the solutions for w* and h*, which G&S call a=x_1 and x_2. They rationalize that one of these is positive and the other negative, and that the positive one must be absent from the final solutions. Looking at the solutions for w* and h* in the subsection prior to the transversality one, it can easily be seen what C[1] must be to make this happen:

```
Solve[91.16095165320871` == 0.00048428934287985255` * C, C]
```

```
{{C → 188237.}}
```

```
Solve[178410.93221097186` == 0.9478017896491786` * C, C]
```

```
{{C → 188237.}}
```

Final Graphics

```
pl75a = Show[plgw1,
  Graphics[Text["Optimal Pumping", {496, 356 * 10^3}]],
  Graphics[Text["Unplanned Pumping", {719, 400 * 10^3}]],
DisplayFunction → Identity];
pl75b = Show[plgw2,
  Graphics[Text["Optimal Elevation", {676, 3.05 * 10^3}]],
  Graphics[Text["Unplanned Elevation", {534, 2.62 * 10^3}]],
DisplayFunction → Identity];
pl75 = Show[GraphicsArray[{{pl75a}, {pl75b}}], ImageSize → 488]
```





- GraphicsArray -